Suppes' Theorem For Probability Logic

Matthew Doty

March 17, 2025

Abstract

We develop finitely additive probability logic and prove a theorem of Patrick Suppes that asserts that $\Psi \vdash \phi$ in classical propositional logic if and only if $(\sum \psi \leftarrow \Psi.\ 1 - \mathcal{P}\psi) \geq 1 - \mathcal{P}\phi$ holds for all probabilities $\mathcal{P}$. We also provide a novel *dual* form of Suppes' Theorem, which holds that $(\sum \phi \leftarrow \Phi.\ \mathcal{P}\phi) \leq \mathcal{P}\psi$ for all probabilities $\mathcal{P}$ if and only $(\bigvee \Phi) \vdash \psi$ and all of the formulae in Φ are logically exclusive from one another. Our proofs use *Maximally Consistent Sets*, and as a consequence, we obtain two *collapse* theorems. In particular, we show $(\sum \phi \leftarrow \Phi.\ \mathcal{P}\phi) \geq \mathcal{P}\psi$ holds for all probabilities $\mathcal{P}$ if and only if $(\sum \phi \leftarrow \Phi.\ \delta\ \phi) \geq \delta\ \psi$ holds for all binary-valued probabilities δ, along with the dual assertion that $(\sum \phi \leftarrow \Phi.\ \mathcal{P}\phi) \leq \mathcal{P}\psi$ holds for all probabilities $\mathcal{P}$ if and only if $(\sum \phi \leftarrow \Phi.\ \delta\ \phi) \leq \delta\ \psi$ holds for all binary-valued probabilities δ.

Contents

Chapter 1

Probability Logic

theory *Probability-Logic*
  imports
    Propositional-Logic-Class.Classical-Connectives
    HOL.Real
    HOL−Library.Countable
begin

unbundle *no funcset-syntax*

1.1 Definition of Probability Logic

Probability logic is defined in terms of an operator over classical logic obeying certain postulates. Scholars often credit George Boole for first conceiving this kind of formulation [1]. Theodore Hailperin in particular has written extensively on this subject [6, 7, 8].

The presentation below roughly follows Kolmogorov's axiomatization [10]. A key difference is that we only require *finite additivity*, rather than *countable additivity*. Finite additivity is also defined in terms of implication ($\rightarrow$).

class *probability-logic* = *classical-logic* +
  fixes $\mathcal{P}$ :: $'a \Rightarrow real$
  assumes *probability-non-negative*: $\mathcal{P}\ \varphi \geq 0$
  assumes *probability-unity*: $\vdash \varphi \Longrightarrow \mathcal{P}\ \varphi = 1$
  assumes *probability-implicational-additivity*:
    $\vdash \varphi \rightarrow \psi \rightarrow \bot \Longrightarrow \mathcal{P}\ ((\varphi \rightarrow \bot) \rightarrow \psi) = \mathcal{P}\ \varphi + \mathcal{P}\ \psi$

A similar axiomatization may be credited to Rescher [11, pg. 185]. However, our formulation has fewer axioms. While Rescher assumes $\vdash \varphi \leftrightarrow \psi \Longrightarrow \mathcal{P}\ \varphi = \mathcal{P}\ \psi$, we show this is a lemma in §1.4.

1.2 Why Finite Additivity?

In this section we touch on why we have chosen to employ finite additivity in our axiomatization of *probability-logic* and deviate from conventional probability theory.

Conventional probability obeys an axiom known as *countable additivity*. Traditionally it states if S is a countable set of sets which are pairwise disjoint, then the limit $\sum\ s \in S.\ \mathcal{P}\ s$ exists and $\mathcal{P}\ (\bigcup\ S) = (\sum\ s \in S.\ \mathcal{P}\ s)$. This is more powerful than our finite additivity axiom $\vdash \varphi \to \psi \to \bot \Longrightarrow \mathcal{P}\ ((\varphi \to \bot) \to \psi) = \mathcal{P}\ \varphi + \mathcal{P}\ \psi$.

However, we argue that demanding countable additivity is not practical.

Historically, the statisticians Bruno de Finetti and Leonard Savage gave the most well known critiques. In [2] de Finetti shows various properties which are true for countably additive probability measures may not hold for finitely additive measures. Savage [12], on the other hand, develops probability based on choices prizes in lotteries.

We instead argue that if we demand countable additivity, then certain properties of real world software would no longer be formally verifiable as we demonstrate here. In particular, it prohibits conventional recursive data structures for defining propositions. Our argument is derivative of one given by Giangiacomo Gerla [5, Section 3].

By taking equivalence classes modulo $\lambda\varphi\ \psi.\ \vdash \varphi \leftrightarrow \psi$, any classical logic instance gives rise to a Boolean algebra known as a *Lindenbaum Algebra*. In the case of *'a classical-propositional-formula* this Boolean algebra algebra is both countable and *atomless*. A theorem of Horn and Tarski [9, Theorem 3.2] asserts there can be no countably additive *Pr* for a countable atomless Boolean algebra.

The above argument is not intended as a blanket refutation of conventional probability theory. It is simply an impossibility result with respect to software implementations of probability logic. Plenty of classic results in probability rely on countable additivity. A nice example, formalized in Isabelle/HOL, is Bouffon's needle [3].

1.3 Basic Properties of Probability Logic

lemma (**in** *probability-logic*) *probability-additivity*:
  assumes $\vdash \sim (\varphi \sqcap \psi)$
  shows $\mathcal{P}\ (\varphi \sqcup \psi) = \mathcal{P}\ \varphi + \mathcal{P}\ \psi$
  using
    assms

unfolding
conjunction-def
disjunction-def
negation-def
by (*simp add*: *probability-implicational-additivity*)

lemma (**in** *probability-logic*) *probability-alternate-additivity*:
assumes $\vdash \varphi \rightarrow \psi \rightarrow \bot$
shows $\mathcal{P}\ (\varphi \sqcup \psi) = \mathcal{P}\ \varphi + \mathcal{P}\ \psi$
using *assms*
by (*metis*
probability-additivity
double-negation-converse
modus-ponens
conjunction-def
negation-def)

lemma (**in** *probability-logic*) *complementation*:
$\mathcal{P}\ (\sim \varphi) = \mathit{1} - \mathcal{P}\ \varphi$
by (*metis*
probability-alternate-additivity
probability-unity
bivalence
negation-elimination
add.commute
add-diff-cancel-left')

lemma (**in** *probability-logic*) *unity-upper-bound*:
$\mathcal{P}\ \varphi \leq \mathit{1}$
by (*metis*
(*no-types*)
diff-ge-0-iff-ge
probability-non-negative
complementation)

1.4 Alternate Definition of Probability Logic

There is an alternate axiomatization of probability logic, due to Brian Gaines [4, pg. 159, postulates P7, P8, and P8] and independently formulated by Brian Weatherson [14]. As Weatherson notes, this axiomatization is suited to formulating *intuitionistic* probability logic. In the case where the underlying logic is classical the Gaines/Weatherson axiomatization is equivalent to the traditional Kolmogorov axiomatization from §1.1.

class *gaines-weatherson-probability* = *classical-logic* +
fixes $\mathcal{P} :: {}'a \Rightarrow \mathit{real}$
assumes *gaines-weatherson-thesis*:
$\mathcal{P}\ \top = \mathit{1}$

assumes *gaines-weatherson-antithesis*:
$\mathcal{P}\ \bot = 0$
assumes *gaines-weatherson-monotonicity*:
$\vdash \varphi \to \psi \Longrightarrow \mathcal{P}\ \varphi \leq \mathcal{P}\ \psi$
assumes *gaines-weatherson-sum-rule*:
$\mathcal{P}\ \varphi + \mathcal{P}\ \psi = \mathcal{P}\ (\varphi \sqcap \psi) + \mathcal{P}\ (\varphi \sqcup \psi)$

sublocale *gaines-weatherson-probability* $\subseteq$ *probability-logic*
proof
fix φ
have $\vdash \bot \to \varphi$
by (*simp add*: *ex-falso-quodlibet*)
thus $0 \leq \mathcal{P}\ \varphi$
using
gaines-weatherson-antithesis
gaines-weatherson-monotonicity
by *fastforce*
next
fix φ
assume $\vdash \varphi$
thus $\mathcal{P}\ \varphi = 1$
by (*metis*
gaines-weatherson-thesis
gaines-weatherson-monotonicity
eq-iff
axiom-k
ex-falso-quodlibet
modus-ponens
verum-def)
next
fix $\varphi\ \psi$
assume $\vdash \varphi \to \psi \to \bot$
hence $\vdash \sim (\varphi \sqcap \psi)$
by (*simp add*: *conjunction-def negation-def*)
thus $\mathcal{P}\ ((\varphi \to \bot) \to \psi) = \mathcal{P}\ \varphi + \mathcal{P}\ \psi$
by (*metis*
add.commute
add.right-neutral
eq-iff
disjunction-def
ex-falso-quodlibet
negation-def
gaines-weatherson-antithesis
gaines-weatherson-monotonicity
gaines-weatherson-sum-rule)
qed

lemma (**in** *probability-logic*) *monotonicity*:
$\vdash \varphi \to \psi \Longrightarrow \mathcal{P}\ \varphi \leq \mathcal{P}\ \psi$

proof −
  assume $\vdash \varphi \to \psi$
  hence $\vdash \sim (\varphi \sqcap \sim \psi)$
    unfolding *negation-def conjunction-def*
    by (*metis*
          conjunction-def
          exclusion-contrapositive-equivalence
          negation-def
          weak-biconditional-weaken)
  hence $\mathcal{P}\ (\varphi \sqcup \sim \psi) = \mathcal{P}\ \varphi + \mathcal{P}\ (\sim \psi)$
    by (*simp add*: *probability-additivity*)
  hence $\mathcal{P}\ \varphi + \mathcal{P}\ (\sim \psi) \leq \mathit{1}$
    by (*metis unity-upper-bound*)
  hence $\mathcal{P}\ \varphi + \mathit{1} - \mathcal{P}\ \psi \leq \mathit{1}$
    by (*simp add*: *complementation*)
  thus *?thesis* **by** *linarith*
qed

lemma (**in** *probability-logic*) *biconditional-equivalence*:
  $\vdash \varphi \leftrightarrow \psi \Longrightarrow \mathcal{P}\ \varphi = \mathcal{P}\ \psi$
  by (*meson*
        eq-iff
        modus-ponens
        biconditional-left-elimination
        biconditional-right-elimination
        monotonicity)

lemma (**in** *probability-logic*) *sum-rule*:
  $\mathcal{P}\ (\varphi \sqcup \psi) + \mathcal{P}\ (\varphi \sqcap \psi) = \mathcal{P}\ \varphi + \mathcal{P}\ \psi$
proof −
  have $\vdash (\varphi \sqcup \psi) \leftrightarrow (\varphi \sqcup \psi \setminus (\varphi \sqcap \psi))$
  proof −
    have $\forall\ \mathfrak{M}.\ \mathfrak{M} \models_{prop} (\langle\varphi\rangle \sqcup \langle\psi\rangle) \leftrightarrow (\langle\varphi\rangle \sqcup \langle\psi\rangle \setminus (\langle\varphi\rangle \sqcap \langle\psi\rangle))$
      unfolding
        classical-logic-class.subtraction-def
        classical-logic-class.negation-def
        classical-logic-class.biconditional-def
        classical-logic-class.conjunction-def
        classical-logic-class.disjunction-def
      by *simp*
    hence $\vdash (\!|\ (\langle\varphi\rangle \sqcup \langle\psi\rangle) \leftrightarrow (\langle\varphi\rangle \sqcup \langle\psi\rangle \setminus (\langle\varphi\rangle \sqcap \langle\psi\rangle))\ |\!)$
      using *propositional-semantics* **by** *blast*
    thus *?thesis* **by** *simp*
  qed
  moreover have $\vdash \varphi \to (\psi \setminus (\varphi \sqcap \psi)) \to \bot$
  proof −
    have $\forall\ \mathfrak{M}.\ \mathfrak{M} \models_{prop} \langle\varphi\rangle \to (\langle\psi\rangle \setminus (\langle\varphi\rangle \sqcap \langle\psi\rangle)) \to \bot$
      unfolding
        classical-logic-class.subtraction-def

classical-logic-class.negation-def
classical-logic-class.biconditional-def
classical-logic-class.conjunction-def
classical-logic-class.disjunction-def
by *simp*
hence $\vdash$ $(\!|$ $\langle\varphi\rangle \rightarrow (\langle\psi\rangle \setminus (\langle\varphi\rangle \sqcap \langle\psi\rangle)) \rightarrow \bot$ $|\!)$
using *propositional-semantics* **by** *blast*
thus *?thesis* **by** *simp*
qed
hence $\mathcal{P}\ (\varphi \sqcup \psi) = \mathcal{P}\ \varphi + \mathcal{P}\ (\psi \setminus (\varphi \sqcap \psi))$
using
probability-alternate-additivity
biconditional-equivalence
calculation
by *auto*
moreover have $\vdash \psi \leftrightarrow (\psi \setminus (\varphi \sqcap \psi) \sqcup (\varphi \sqcap \psi))$
proof $-$
have $\forall\ \mathfrak{M}.\ \mathfrak{M} \models_{prop} \langle\psi\rangle \leftrightarrow (\langle\psi\rangle \setminus (\langle\varphi\rangle \sqcap \langle\psi\rangle) \sqcup (\langle\varphi\rangle \sqcap \langle\psi\rangle))$
unfolding
classical-logic-class.subtraction-def
classical-logic-class.negation-def
classical-logic-class.biconditional-def
classical-logic-class.conjunction-def
classical-logic-class.disjunction-def
by *auto*
hence $\vdash$ $(\!|$ $\langle\psi\rangle \leftrightarrow (\langle\psi\rangle \setminus (\langle\varphi\rangle \sqcap \langle\psi\rangle) \sqcup (\langle\varphi\rangle \sqcap \langle\psi\rangle))$ $|\!)$
using *propositional-semantics* **by**
blast
thus *?thesis* **by** *simp*
qed
moreover have $\vdash (\psi \setminus (\varphi \sqcap \psi)) \rightarrow (\varphi \sqcap \psi) \rightarrow \bot$
unfolding
subtraction-def
negation-def
conjunction-def
using
conjunction-def
conjunction-right-elimination
by *auto*
hence $\mathcal{P}\ \psi = \mathcal{P}\ (\psi \setminus (\varphi \sqcap \psi)) + \mathcal{P}\ (\varphi \sqcap \psi)$
using
probability-alternate-additivity
biconditional-equivalence
calculation
by *auto*
ultimately show *?thesis*
by *simp*
qed

sublocale *probability-logic* $\subseteq$ *gaines-weatherson-probability*
proof
  show $\mathcal{P}\ \top = 1$
    by (*simp add*: *probability-unity*)
next
  show $\mathcal{P}\ \bot = 0$
    by (*metis*
        add-cancel-left-right
        probability-additivity
        ex-falso-quodlibet
        probability-unity
        bivalence
        conjunction-right-elimination
        negation-def)
next
  fix $\varphi\ \psi$
  assume $\vdash \varphi \to \psi$
  thus $\mathcal{P}\ \varphi \leq \mathcal{P}\ \psi$
    using *monotonicity*
    by *auto*
next
  fix $\varphi\ \psi$
  show $\mathcal{P}\ \varphi + \mathcal{P}\ \psi = \mathcal{P}\ (\varphi \sqcap \psi) + \mathcal{P}\ (\varphi \sqcup \psi)$
    by (*metis sum-rule add.commute*)
qed

sublocale *probability-logic* $\subseteq$ *consistent-classical-logic*
proof
  show $\neg \vdash \bot$ **using** *probability-unity gaines-weatherson-antithesis* **by** *auto*
qed

lemma (**in** *probability-logic*) *subtraction-identity*:
  $\mathcal{P}\ (\varphi \setminus \psi) = \mathcal{P}\ \varphi - \mathcal{P}\ (\varphi \sqcap \psi)$
proof –
  have $\vdash \varphi \leftrightarrow ((\varphi \setminus \psi) \sqcup (\varphi \sqcap \psi))$
  proof –
    have $\forall\ \mathfrak{M}.\ \mathfrak{M} \models_{prop} \langle\varphi\rangle \leftrightarrow (((\langle\varphi\rangle \setminus \langle\psi\rangle) \sqcup (\langle\varphi\rangle \sqcap \langle\psi\rangle))$
      unfolding
        classical-logic-class.subtraction-def
        classical-logic-class.negation-def
        classical-logic-class.biconditional-def
        classical-logic-class.conjunction-def
        classical-logic-class.disjunction-def
      by (*simp*, *blast*)
    hence $\vdash (\!|\ \langle\varphi\rangle \leftrightarrow ((\langle\varphi\rangle \setminus \langle\psi\rangle) \sqcup (\langle\varphi\rangle \sqcap \langle\psi\rangle))\ |\!)$
      using *propositional-semantics* **by** *blast*
    thus *?thesis* **by** *simp*
  qed
  hence $\mathcal{P}\ \varphi\ = \mathcal{P}\ ((\varphi \setminus \psi) \sqcup (\varphi \sqcap \psi))$

using *biconditional-equivalence*
by *simp*
moreover have $\vdash \sim((\varphi \setminus \psi) \sqcap (\varphi \sqcap \psi))$
proof $-$
have $\forall\ \mathfrak{M}.\ \mathfrak{M} \models_{prop} \sim((\langle\varphi\rangle \setminus \langle\psi\rangle) \sqcap (\langle\varphi\rangle \sqcap \langle\psi\rangle))$
unfolding
classical-logic-class.subtraction-def
classical-logic-class.negation-def
classical-logic-class.conjunction-def
classical-logic-class.disjunction-def
by *simp*
hence $\vdash (\!|\ \sim((\langle\varphi\rangle \setminus \langle\psi\rangle) \sqcap (\langle\varphi\rangle \sqcap \langle\psi\rangle))\ |\!)$
using *propositional-semantics* **by** *blast*
thus *?thesis* **by** *simp*
qed
ultimately show *?thesis*
using *probability-additivity*
by *auto*
qed

1.5 Basic Probability Logic Inequality Results

lemma (**in** *probability-logic*) *disjunction-sum-inequality*:
$\mathcal{P}\ (\varphi \sqcup \psi) \leq \mathcal{P}\ \varphi + \mathcal{P}\ \psi$
proof $-$
have $\mathcal{P}\ (\varphi \sqcup \psi) + \mathcal{P}\ (\varphi \sqcap \psi) = \mathcal{P}\ \varphi + \mathcal{P}\ \psi$
$0 \leq \mathcal{P}\ (\varphi \sqcap \psi)$
by (*simp add*: *sum-rule*, *simp add*: *probability-non-negative*)
thus *?thesis* **by** *linarith*
qed

lemma (**in** *probability-logic*)
arbitrary-disjunction-list-summation-inequality:
$\mathcal{P}\ (\bigsqcup\ \Phi) \leq (\sum \varphi \leftarrow \Phi.\ \mathcal{P}\ \varphi)$
proof (*induct* Φ)
case *Nil*
then show *?case* **by** (*simp add*: *gaines-weatherson-antithesis*)
next
case (*Cons* φ Φ)
have $\mathcal{P}\ (\bigsqcup\ (\varphi\ \#\ \Phi)) \leq \mathcal{P}\ \varphi + \mathcal{P}\ (\bigsqcup\ \Phi)$
using *disjunction-sum-inequality*
by *simp*
with *Cons* **have** $\mathcal{P}\ (\bigsqcup\ (\varphi\ \#\ \Phi)) \leq \mathcal{P}\ \varphi + (\sum \varphi \leftarrow \Phi.\ \mathcal{P}\ \varphi)$ **by** *linarith*
then show *?case* **by** *simp*
qed

lemma (**in** *probability-logic*) *implication-list-summation-inequality*:
assumes $\vdash \varphi \to \bigsqcup\ \Psi$
shows $\mathcal{P}\ \varphi \leq (\sum \psi \leftarrow \Psi.\ \mathcal{P}\ \psi)$

using
assms
arbitrary-disjunction-list-summation-inequality
monotonicity
order-trans
by *blast*

lemma (**in** *probability-logic*) *arbitrary-disjunction-set-summation-inequality*:
$\mathcal{P}$ ($\bigsqcup$ Φ) $\leq$ ($\sum \varphi \in$ *set* Φ. $\mathcal{P}$ φ)
by (*metis*
arbitrary-disjunction-list-summation-inequality
arbitrary-disjunction-remdups
biconditional-equivalence
sum.set-conv-list)

lemma (**in** *probability-logic*) *implication-set-summation-inequality*:
assumes $\vdash \varphi \rightarrow \bigsqcup \Psi$
shows $\mathcal{P}$ $\varphi \leq$ ($\sum \psi \in$ *set* Ψ. $\mathcal{P}$ ψ)
using
assms
arbitrary-disjunction-set-summation-inequality
monotonicity
order-trans
by *blast*

1.6 Dirac Measures

Before presenting *Dirac measures* in probability logic, we first give the set of all functions satisfying probability logic.

definition (**in** *classical-logic*) *probabilities* :: ($'a \Rightarrow$ *real*) *set*
where *probabilities* =
$\{\mathcal{P}$. *class.probability-logic* (λ φ. $\vdash \varphi$) ($\rightarrow$) $\bot$ $\mathcal{P}$ $\}$

Traditionally, a Dirac measure is a function δ_x where $\delta_x\ S = 1$ if $x \in S$ and $\delta_x\ S = 0$ otherwise. This means that Dirac measures correspond to special ultrafilters on their underlying σ-algebra which are closed under countable unions.

Probability logic, as discussed in §1.2, may not have countable joins in its underlying logic. In the setting of probability logic, Dirac measures are simple probability functions that are either 0 or 1.

definition (**in** *classical-logic*) *dirac-measures* :: ($'a \Rightarrow$ *real*) *set*
where *dirac-measures* =
$\{$ $\mathcal{P}$. *class.probability-logic* (λ φ. $\vdash \varphi$) ($\rightarrow$) $\bot$ $\mathcal{P}$
$\wedge$ ($\forall x$. $\mathcal{P}$ $x = 0 \vee \mathcal{P}$ $x = 1$) $\}$

lemma (**in** *classical-logic*) *dirac-measures-subset*:

```
  dirac-measures ⊆ probabilities
  unfolding
    probabilities-def
    dirac-measures-def
  by fastforce
```

Maximally consistent sets correspond to Dirac measures. One direction of this correspondence is established below.

```
lemma (in classical-logic) MCS-dirac-measure:
  assumes MCS Ω
    shows (λ χ. if χ∈Ω then (1 :: real) else 0) ∈ dirac-measures
      (is ?𝒫 ∈ dirac-measures)
proof −
  have class.probability-logic (λ φ. ⊢ φ) (→) ⊥ ?𝒫
  proof (standard, simp,
         meson
           assms
           formula-maximally-consistent-set-def-reflection
           maximally-consistent-set-def
           set-deduction-weaken)
    fix φ ψ
    assume ⊢ φ → ψ → ⊥
    hence φ ⊓ ψ ∉ Ω
      by (metis
            assms
            formula-consistent-def
            formula-maximally-consistent-set-def-def
            maximally-consistent-set-def
            conjunction-def
            set-deduction-modus-ponens
            set-deduction-reflection
            set-deduction-weaken)
    hence φ ∉ Ω ∨ ψ ∉ Ω
      using
        assms
        formula-maximally-consistent-set-def-reflection
        maximally-consistent-set-def
        conjunction-set-deduction-equivalence
      by meson
    have φ ⊔ ψ ∈ Ω = (φ ∈ Ω ∨ ψ ∈ Ω)
      by (metis
            ‹φ ⊓ ψ ∉ Ω›
            assms
            formula-maximally-consistent-set-def-implication
            maximally-consistent-set-def
            conjunction-def
            disjunction-def)
    have ?𝒫 (φ ⊔ ψ) = ?𝒫 φ + ?𝒫 ψ
    proof (cases φ ⊔ ψ ∈ Ω)
```

```
      case True
      hence ◇: 1 = ?𝒫 (φ ⊔ ψ) by simp
      show ?thesis
      proof (cases φ ∈ Ω)
        case True
        hence ψ ∉ Ω
          using ‹φ ∉ Ω ∨ ψ ∉ Ω›
          by blast
        have ?𝒫 (φ ⊔ ψ) = (1::real) using ◇ by simp
        also have ... = 1 + (0::real) by linarith
        also have ... = ?𝒫 φ + ?𝒫 ψ
          using ‹ψ ∉ Ω› ‹φ ∈ Ω› by simp
        finally show ?thesis .
      next
        case False
        hence ψ ∈ Ω
          using ‹φ ⊔ ψ ∈ Ω› ‹(φ ⊔ ψ ∈ Ω) = (φ ∈ Ω ∨ ψ ∈ Ω)›
          by blast
        have ?𝒫 (φ ⊔ ψ) = (1::real) using ◇ by simp
        also have ... = (0::real) + 1 by linarith
        also have ... = ?𝒫 φ + ?𝒫 ψ
          using ‹ψ ∈ Ω› ‹φ ∉ Ω› by simp
        finally show ?thesis .
      qed
    next
      case False
      moreover from this have φ ∉ Ω ψ ∉ Ω
        using ‹(φ ⊔ ψ ∈ Ω) = (φ ∈ Ω ∨ ψ ∈ Ω)› by blast+
      ultimately show ?thesis by simp
    qed
    thus ?𝒫 ((φ → ⊥) → ψ) = ?𝒫 φ + ?𝒫 ψ
      unfolding disjunction-def .
  qed
  thus ?thesis
    unfolding dirac-measures-def
    by simp
qed

unbundle funcset-syntax

end
```

Chapter 2

Suppes' Theorem

theory *Suppes-Theorem*
  imports *Probability-Logic*
begin

unbundle *no funcset-syntax*

An elementary completeness theorem for inequalities for probability logic is due to Patrick Suppes [13].

A consequence of this Suppes' theorem is an elementary form of *collapse*, which asserts that inequalities for probabilities are logically equivalent to the more restricted class of *Dirac measures* as defined in §1.6.

2.1 Suppes' List Theorem

We first establish Suppes' theorem for lists of propositions. This is done by establishing our first completeness theorem using *Dirac measures*.

First, we use the result from §1.5 that shows $\vdash \varphi \to \bigsqcup\ \Psi$ implies $\mathcal{P}\ \varphi \le (\sum \psi{\leftarrow}\Psi.\ \mathcal{P}\ \psi)$. This can be understood as a *soundness* result.

To show completeness, assume $\neg \vdash \varphi \to \bigsqcup\ \Psi$. From this obtain a maximally consistent Ω such that $\varphi \to \bigsqcup\ \Psi \notin \Omega$. We then define $\delta\ \chi = (\mathit{if}\ \chi \in \Omega\ \mathit{then}\ 1\ \mathit{else}\ 0)$ and show δ is a *Dirac measure* such that $\delta\ \varphi \le (\sum \psi{\leftarrow}\Psi.\ \delta\ \psi)$.

lemma (**in** *classical-logic*) *dirac-list-summation-completeness*:
  $(\forall\ \delta \in \mathit{dirac\text{-}measures}.\ \delta\ \varphi \le (\sum \psi{\leftarrow}\Psi.\ \delta\ \psi)) = \vdash \varphi \to \bigsqcup\ \Psi$
proof −
  {
    fix $\delta :: {'a} \Rightarrow \mathit{real}$
    assume $\delta \in \mathit{dirac\text{-}measures}$
    from *this* **interpret** *probability-logic* $(\lambda\ \varphi.\ \vdash \varphi)\ (\to)\ \bot\ \delta$
      unfolding *dirac-measures-def*

by *auto*
assume $\vdash \varphi \rightarrow \bigsqcup \Psi$
hence $\delta\ \varphi \leq (\sum \psi \leftarrow \Psi.\ \delta\ \psi)$
using *implication-list-summation-inequality*
by *auto*
}
moreover {
assume $\neg \vdash \varphi \rightarrow \bigsqcup \Psi$
from *this* **obtain** Ω **where** Ω:
MCS Ω
$\varphi \in \Omega$
$\bigsqcup \Psi \notin \Omega$
by (*meson*
insert-subset
formula-consistent-def
formula-maximal-consistency
formula-maximally-consistent-extension
formula-maximally-consistent-set-def-def
set-deduction-base-theory
set-deduction-reflection
set-deduction-theorem)
hence $\forall\ \psi \in$ *set* $\Psi.\ \psi \notin \Omega$
using *arbitrary-disjunction-exclusion-MCS* **by** *blast*
define δ **where** $\delta = (\lambda\ \chi\ .\ \mathit{if}\ \chi{\in}\Omega\ \mathit{then}\ (\mathit{1} :: \mathit{real})\ \mathit{else}\ \mathit{0})$
from ‹$\forall\ \psi \in$ *set* $\Psi.\ \psi \notin \Omega$› **have** $(\sum \psi \leftarrow \Psi.\ \delta\ \psi) = \mathit{0}$
unfolding δ*-def*
by (*induct* Ψ, *simp*, *simp*)
hence $\neg\ \delta\ \varphi \leq (\sum \psi \leftarrow \Psi.\ \delta\ \psi)$
unfolding δ*-def*
by (*simp add*: $\Omega(\mathit{2})$)
hence
$\exists\ \delta \in$ *dirac-measures*. $\neg\ (\delta\ \varphi \leq (\sum \psi \leftarrow \Psi.\ \delta\ \psi))$
unfolding δ*-def*
using $\Omega(\mathit{1})$ *MCS-dirac-measure* **by** *auto*
}
ultimately show *?thesis* **by** *blast*
qed

theorem (**in** *classical-logic*) *list-summation-completeness*:
$(\forall\ \mathcal{P} \in$ *probabilities*. $\mathcal{P}\ \varphi \leq (\sum \psi \leftarrow \Psi.\ \mathcal{P}\ \psi)) = \vdash \varphi \rightarrow \bigsqcup \Psi$
(**is** *?lhs* = *?rhs*)
proof
assume *?lhs*
hence $\forall\ \delta \in$ *dirac-measures*. $\delta\ \varphi \leq (\sum \psi \leftarrow \Psi.\ \delta\ \psi)$
unfolding *dirac-measures-def probabilities-def*
by *blast*
thus *?rhs*
using *dirac-list-summation-completeness* **by** *blast*
next

```
  assume ?rhs
  show ?lhs
  proof
    fix 𝒫 :: 'a ⇒ real
    assume 𝒫 ∈ probabilities
    from this interpret probability-logic (λ φ. ⊢ φ) (→) ⊥ 𝒫
      unfolding probabilities-def
      by auto
    show 𝒫 φ ≤ (∑ψ←Ψ. 𝒫 ψ)
      using ‹?rhs› implication-list-summation-inequality
      by simp
  qed
qed
```

The collapse theorem asserts that to prove an inequalities for all probabilities in probability logic, one only needs to consider the case of functions which take on values of 0 or 1.

```
lemma (in classical-logic) suppes-collapse:
  (∀ 𝒫 ∈ probabilities. 𝒫 φ ≤ (∑ψ←Ψ. 𝒫 ψ))
     = (∀ δ ∈ dirac-measures. δ φ ≤ (∑ψ←Ψ. δ ψ))
  by (simp add:
        dirac-list-summation-completeness
        list-summation-completeness)

lemma (in classical-logic) probability-member-neg:
  fixes 𝒫
  assumes 𝒫 ∈ probabilities
  shows 𝒫 (∼ φ) = 1 − 𝒫 φ
proof −
  from assms interpret probability-logic (λ φ. ⊢ φ) (→) ⊥ 𝒫
    unfolding probabilities-def
    by auto
  show ?thesis
    by (simp add: complementation)
qed
```

Suppes' theorem has a philosophical interpretation. It asserts that if $\Psi :\vdash \varphi$, then our *uncertainty* in φ is bounded above by our uncertainty in Ψ. Here the uncertainty in the proposition φ is $1 - \mathcal{P}\ \varphi$. Our uncertainty in Ψ, on the other hand, is $\sum \psi \leftarrow \Psi.\ 1 - \mathcal{P}\ \psi$.

```
theorem (in classical-logic) suppes-list-theorem:
  Ψ :⊢ φ = (∀ 𝒫 ∈ probabilities. (∑ψ←Ψ. 1 − 𝒫 ψ) ≥ 1 − 𝒫 φ)
proof −
  have
    Ψ :⊢ φ = (∀ 𝒫 ∈ probabilities. (∑ψ ← ∼ Ψ. 𝒫 ψ) ≥ 𝒫 (∼ φ))
    using
      list-summation-completeness
      weak-biconditional-weaken
```

contra-list-curry-uncurry
list-deduction-def
by *blast*
moreover have
$\forall\ \mathcal{P} \in$ *probabilities*. $(\sum \psi \leftarrow (\sim \Psi).\ \mathcal{P}\ \psi) = (\sum \psi \leftarrow \Psi.\ \mathcal{P}\ (\sim \psi))$
by (*induct* Ψ, *auto*)
ultimately show *?thesis*
using *probability-member-neg*
by (*induct* Ψ, *simp+*)
qed

2.2 Suppes' Set Theorem

Suppes theorem also obtains for *sets*.

lemma (**in** *classical-logic*) *dirac-set-summation-completeness*:
$(\forall\ \delta \in$ *dirac-measures*. $\delta\ \varphi \leq (\sum \psi \in$ *set* $\Psi.\ \delta\ \psi)) = \vdash \varphi \rightarrow \bigsqcup\ \Psi$
by (*metis*
dirac-list-summation-completeness
modus-ponens
arbitrary-disjunction-remdups
biconditional-left-elimination
biconditional-right-elimination
hypothetical-syllogism
sum.set-conv-list)

theorem (**in** *classical-logic*) *set-summation-completeness*:
$(\forall\ \delta \in$ *probabilities*. $\delta\ \varphi \leq (\sum \psi \in$ *set* $\Psi.\ \delta\ \psi)) = \vdash \varphi \rightarrow \bigsqcup\ \Psi$
by (*metis*
dirac-list-summation-completeness
dirac-set-summation-completeness
list-summation-completeness
sum.set-conv-list)

lemma (**in** *classical-logic*) *suppes-set-collapse*:
$(\forall\ \mathcal{P} \in$ *probabilities*. $\mathcal{P}\ \varphi \leq (\sum \psi \in$ *set* $\Psi.\ \mathcal{P}\ \psi))$
$= (\forall\ \delta \in$ *dirac-measures*. $\delta\ \varphi \leq (\sum \psi \in$ *set* $\Psi.\ \delta\ \psi))$
by (*simp add*:
dirac-set-summation-completeness
set-summation-completeness)

In our formulation of logic, there is not reason that $\sim a = \sim b$ while $a \neq b$. As a consequence the Suppes theorem for sets presented below is different than the one given in §2.1.

theorem (**in** *classical-logic*) *suppes-set-theorem*:
$\Psi :\vdash \varphi$
$= (\forall\ \mathcal{P} \in$ *probabilities*. $(\sum \psi \in$ *set* $(\sim \Psi).\ \mathcal{P}\ \psi) \geq 1 - \mathcal{P}\ \varphi)$
proof −
have $\Psi :\vdash \varphi$

$= (\forall\ \mathcal{P} \in \textit{probabilities}.\ (\sum\psi \in \textit{set}\ (\sim\ \Psi).\ \mathcal{P}\ \psi) \geq \mathcal{P}\ (\sim\varphi))$
using
contra-list-curry-uncurry
list-deduction-def
set-summation-completeness
weak-biconditional-weaken
by *blast*
thus *?thesis*
using *probability-member-neg*
by (*induct* Ψ, *auto*)
qed

2.3 Converse Suppes' Theorem

A formulation of the converse of Suppes' theorem obtains for lists/sets of *logically disjoint* propositions.

lemma (**in** *probability-logic*) *exclusive-sum-list-identity*:
assumes $\vdash \coprod\ \Phi$
shows $\mathcal{P}\ (\bigsqcup\ \Phi) = (\sum\varphi\leftarrow\Phi.\ \mathcal{P}\ \varphi)$
using *assms*
proof (*induct* Φ)
case *Nil*
then show *?case*
by (*simp add*: *gaines-weatherson-antithesis*)
next
case (*Cons* φ Φ)
assume $\vdash \coprod\ (\varphi\ \#\ \Phi)$
hence $\vdash\ \sim(\varphi \sqcap \bigsqcup\ \Phi) \vdash \coprod\ \Phi$ **by** *simp+*
hence $\mathcal{P}\ (\bigsqcup(\varphi\ \#\ \Phi)) = \mathcal{P}\ \varphi + \mathcal{P}\ (\bigsqcup\ \Phi)$
$\mathcal{P}\ (\bigsqcup\ \Phi) = (\sum\varphi\leftarrow\Phi.\ \mathcal{P}\ \varphi)$
using *Cons.hyps probability-additivity* **by** *auto*
hence $\mathcal{P}\ (\bigsqcup(\varphi\ \#\ \Phi)) = \mathcal{P}\ \varphi + (\sum\varphi\leftarrow\Phi.\ \mathcal{P}\ \varphi)$ **by** *auto*
thus *?case* **by** *simp*
qed

lemma *sum-list-monotone*:
fixes $f :: 'a \Rightarrow \textit{real}$
assumes $\forall\ x.\ f\, x \geq 0$
and $\textit{set}\ \Phi \subseteq \textit{set}\ \Psi$
and *distinct* Φ
shows $(\sum\varphi\leftarrow\Phi.\ f\ \varphi) \leq (\sum\psi\leftarrow\Psi.\ f\ \psi)$
using *assms*
proof −
assume $\forall\ x.\ f\, x \geq 0$
have $\forall\Phi.\ \textit{set}\ \Phi \subseteq \textit{set}\ \Psi$
$\longrightarrow$ *distinct* Φ
$\longrightarrow (\sum\varphi\leftarrow\Phi.\ f\ \varphi) \leq (\sum\psi\leftarrow\Psi.\ f\ \psi)$
proof (*induct* Ψ)

case *Nil*
then show *?case* **by** *simp*
next
case (*Cons* ψ Ψ)
{
fix Φ
assume *set* $\Phi \subseteq$ *set* $(\psi \# \Psi)$
and *distinct* Φ
have $(\sum\varphi\leftarrow\Phi.\ f\ \varphi) \leq (\sum\psi'\leftarrow(\psi \# \Psi).\ f\ \psi')$
proof $-$
{
assume $\psi \notin$ *set* Φ
with ‹*set* $\Phi \subseteq$ *set* $(\psi \# \Psi)$› **have** *set* $\Phi \subseteq$ *set* Ψ **by** *auto*
hence $(\sum\varphi\leftarrow\Phi.\ f\ \varphi) \leq (\sum\psi\leftarrow\Psi.\ f\ \psi)$
using *Cons.hyps* ‹*distinct* Φ› **by** *auto*
moreover have $f\ \psi \geq 0$ **using** ‹$\forall\ x.\ f\ x \geq 0$› **by** *metis*
ultimately have *?thesis* **by** *simp*
}
moreover
{
assume $\psi \in$ *set* Φ
hence *set* $\Phi =$ *insert* ψ (*set* (*removeAll* ψ Φ))
by *auto*
with ‹*set* $\Phi \subseteq$ *set* $(\psi \# \Psi)$› **have** *set* (*removeAll* ψ Φ) $\subseteq$ *set* Ψ
by (*metis*
insert-subset
list.simps(*15*)
set-removeAll
subset-insert-iff)
moreover from ‹*distinct* Φ› **have** *distinct* (*removeAll* ψ Φ)
by (*meson distinct-removeAll*)
ultimately have $(\sum\varphi\leftarrow(\mathit{removeAll}\ \psi\ \Phi).\ f\ \varphi) \leq (\sum\psi\leftarrow\Psi.\ f\ \psi)$
using *Cons.hyps*
by *simp*
moreover from ‹$\psi \in$ *set* Φ› ‹*distinct* Φ›
have $(\sum\varphi\leftarrow\Phi.\ f\ \varphi) = f\ \psi + (\sum\varphi\leftarrow(\mathit{removeAll}\ \psi\ \Phi).\ f\ \varphi)$
using *distinct-remove1-removeAll sum-list-map-remove1*
by *fastforce*
ultimately have *?thesis* **using** ‹$\forall\ x.\ f\ x \geq 0$›
by *simp*
}
ultimately show *?thesis* **by** *blast*
qed
}
thus *?case* **by** *blast*
qed
moreover assume *set* $\Phi \subseteq$ *set* Ψ **and** *distinct* Φ
ultimately show *?thesis* **by** *blast*
qed

lemma *count-remove-all-sum-list*:
  fixes f :: $'a \Rightarrow$ *real*
  shows *real* (*count-list xs x*) $*$ $f\ x$ + ($\sum x' \leftarrow$(*removeAll x xs*). $f\ x'$)
   = ($\sum x \leftarrow xs.\ f\ x$)
  by (*induct xs*, *simp*, *simp*, *metis combine-common-factor mult-cancel-right1*)

lemma (**in** *classical-logic*) *dirac-exclusive-implication-completeness*:
  ($\forall\ \delta \in$ *dirac-measures*. ($\sum \varphi \leftarrow \Phi.\ \delta\ \varphi$) $\leq \delta\ \psi$) = ($\vdash \coprod\ \Phi\ \wedge\ \vdash \bigsqcup\ \Phi \rightarrow \psi$)
proof −
  {
   fix δ
   assume $\delta \in$ *dirac-measures*
   from *this* **interpret** *probability-logic* ($\lambda\ \varphi.\ \vdash \varphi$) ($\rightarrow$) $\perp$ δ
    unfolding *dirac-measures-def*
    by *simp*
   assume $\vdash \coprod\ \Phi \vdash \bigsqcup\ \Phi \rightarrow \psi$
   hence ($\sum \varphi \leftarrow \Phi.\ \delta\ \varphi$) $\leq \delta\ \psi$
    using *exclusive-sum-list-identity monotonicity* **by** *fastforce*
  }
  moreover
  {
   assume $\neg \vdash \coprod\ \Phi$
   hence ($\exists\ \varphi \in$ *set* Φ. $\exists\ \psi \in$ *set* Φ.
     $\varphi \neq \psi \wedge \neg \vdash \sim (\varphi \sqcap \psi)$) $\vee$ ($\exists\ \varphi \in$ *duplicates* Φ. $\neg \vdash \sim \varphi$)
    using *exclusive-equivalence set-deduction-base-theory* **by** *blast*
   hence $\neg$ ($\forall\ \delta \in$ *dirac-measures*. ($\sum \varphi \leftarrow \Phi.\ \delta\ \varphi$) $\leq \delta\ \psi$)
   proof (*elim disjE*)
    assume $\exists\ \varphi \in$ *set* Φ. $\exists\ \chi \in$ *set* Φ. $\varphi \neq \chi \wedge \neg \vdash \sim (\varphi \sqcap \chi)$
    from *this* **obtain** φ **and** χ
     where *φχ-properties*:
      $\varphi \in$ *set* Φ
      $\chi \in$ *set* Φ
      $\varphi \neq \chi$
      $\neg \vdash \sim (\varphi \sqcap \chi)$
     by *blast*
    from *this* **obtain** Ω **where** Ω: *MCS* Ω $\sim (\varphi \sqcap \chi) \notin \Omega$
     by (*meson*
       insert-subset
       formula-consistent-def
       formula-maximal-consistency
       formula-maximally-consistent-extension
       formula-maximally-consistent-set-def-def
       set-deduction-base-theory
       set-deduction-reflection
       set-deduction-theorem)
    let *?δ* = $\lambda\ \chi$. *if* $\chi \in \Omega$ *then* (*1* :: *real*) *else 0*
    from Ω **have** $\varphi \in \Omega\ \chi \in \Omega$
     by (*metis*

```
          formula-maximally-consistent-set-def-implication
          maximally-consistent-set-def
          conjunction-def
          negation-def)+
  with φχ-properties have
      (∑ φ←[φ, χ]. ?δ φ) = 2
      set [φ, χ] ⊆ set Φ
      distinct [φ, χ]
      ∀ φ. ?δ φ ≥ 0
    by simp+
  hence (∑ φ←Φ. ?δ φ) ≥ 2 using sum-list-monotone by metis
  hence ¬ (∑ φ←Φ. ?δ φ) ≤ ?δ (ψ) by auto
  thus ?thesis
    using Ω(1) MCS-dirac-measure
    by auto
next
  assume ∃ φ ∈ duplicates Φ. ¬ ⊢ ∼ φ
  from this obtain φ where φ: φ ∈ duplicates Φ ¬ ⊢ ∼ φ
    using
      exclusive-equivalence [where Γ={}]
      set-deduction-base-theory
    by blast
  from φ obtain Ω where Ω: MCS Ω ∼ φ ∉ Ω
    by (meson
          insert-subset
          formula-consistent-def
          formula-maximal-consistency
          formula-maximally-consistent-extension
          formula-maximally-consistent-set-def-def
          set-deduction-base-theory
          set-deduction-reflection
          set-deduction-theorem)
  hence φ ∈ Ω
    using negation-def by auto
  let ?δ = λ χ. if χ∈Ω then (1 :: real) else 0
  from φ have count-list Φ φ ≥ 2
    using duplicates-alt-def [where xs=Φ]
    by blast
  hence real (count-list Φ φ) * ?δ φ ≥ 2 using ‹φ ∈ Ω› by simp
  moreover
  {
    fix Ψ
    have (∑ φ←Ψ. ?δ φ) ≥ 0 by (induct Ψ, simp, simp)
  }
  moreover have (0::real)
                        ≤ (∑ a←removeAll φ Φ. if a ∈ Ω then 1 else 0)
    using ‹⋀Ψ. 0 ≤ (∑ φ←Ψ. if φ ∈ Ω then 1 else 0)›
    by presburger
  ultimately have real (count-list Φ φ) * ?δ φ
```

$+ (\sum \ \varphi \leftarrow (\mathit{removeAll}\ \varphi\ \Phi).\ \mathit{?\delta}\ \varphi) \geq \mathit{2}$
using ‹$\mathit{2} \leq \mathit{real}\ (\mathit{count\text{-}list}\ \Phi\ \varphi) * (\mathit{if}\ \varphi \in \Omega\ \mathit{then}\ \mathit{1}\ \mathit{else}\ \mathit{0})$›
by *linarith*
hence $(\sum \varphi \leftarrow \Phi.\ \mathit{?\delta}\ \varphi) \geq \mathit{2}$ **by** (*metis count-remove-all-sum-list*)
hence $\neg\ (\sum \varphi \leftarrow \Phi.\ \mathit{?\delta}\ \varphi) \leq \mathit{?\delta}\ (\psi)$ **by** *auto*
thus *?thesis*
using $\Omega(\mathit{1})$ *MCS-dirac-measure*
by *auto*
qed
}
moreover
{
assume $\neg \vdash \bigsqcup\ \Phi \rightarrow \psi$
from *this* **obtain** $\Omega\ \varphi$
where
Ω: $\mathit{MCS}\ \Omega$
and ψ: $\psi \notin \Omega$
and φ: $\varphi \in \mathit{set}\ \Phi\ \varphi \in \Omega$
by (*meson*
insert-subset
formula-consistent-def
formula-maximal-consistency
formula-maximally-consistent-extension
formula-maximally-consistent-set-def-def
arbitrary-disjunction-exclusion-MCS
set-deduction-base-theory
set-deduction-reflection
set-deduction-theorem)
let $\mathit{?\delta} = \lambda\ \chi.\ \mathit{if}\ \chi{\in}\Omega\ \mathit{then}\ (\mathit{1} :: \mathit{real})\ \mathit{else}\ \mathit{0}$
from φ **have** $(\sum \varphi \leftarrow \Phi.\ \mathit{?\delta}\ \varphi) \geq \mathit{1}$
proof (*induct* Φ)
case *Nil*
then show *?case* **by** *simp*
next
case $(\mathit{Cons}\ \varphi'\ \Phi)$
obtain $f :: \mathit{real}\ \mathit{list} \Rightarrow \mathit{real}$ **where** f:
$\forall\, rs.\ f\ rs \in \mathit{set}\ rs \wedge \neg\ \mathit{0} \leq f\ rs \vee \mathit{0} \leq \mathit{sum\text{-}list}\ rs$
using *sum-list-nonneg* **by** *metis*
moreover have $f\ (\mathit{map}\ \mathit{?\delta}\ \Phi) \notin \mathit{set}\ (\mathit{map}\ \mathit{?\delta}\ \Phi) \vee \mathit{0} \leq f\ (\mathit{map}\ \mathit{?\delta}\ \Phi)$
by *fastforce*
ultimately show *?case*
by (*simp*, *metis Cons.hyps Cons.prems*$(\mathit{1})$ $\varphi(\mathit{2})$ *set-ConsD*)
qed
hence $\neg\ (\sum \varphi \leftarrow \Phi.\ \mathit{?\delta}\ \varphi) \leq \mathit{?\delta}\ (\psi)$ **using** ψ **by** *auto*
hence $\neg\ (\forall\ \delta \in \mathit{dirac\text{-}measures}.\ (\sum \varphi \leftarrow \Phi.\ \delta\ \varphi) \leq \delta\ \psi)$
using $\Omega(\mathit{1})$ *MCS-dirac-measure*
by *auto*
}
ultimately show *?thesis* **by** *blast*

qed

theorem (**in** *classical-logic*) *exclusive-implication-completeness*:
  $(\forall\ \mathcal{P} \in probabilities.\ (\sum \varphi \leftarrow \Phi.\ \mathcal{P}\ \varphi) \leq \mathcal{P}\ \psi) = (\vdash \coprod\ \Phi \wedge\ \vdash \bigsqcup\ \Phi \to \psi)$
  (**is** *?lhs* = *?rhs*)
proof
  assume *?lhs*
  thus *?rhs*
    by (*meson*
        dirac-exclusive-implication-completeness
        dirac-measures-subset
        subset-eq)
next
  assume *?rhs*
  show *?lhs*
  proof
    fix $\mathcal{P}$ :: $'a \Rightarrow real$
    assume $\mathcal{P} \in probabilities$
    from *this* **interpret** *probability-logic* $(\lambda\ \varphi.\ \vdash \varphi)$ $(\to)$ $\perp$ $\mathcal{P}$
      unfolding *probabilities-def*
      by *simp*
    show $(\sum \varphi \leftarrow \Phi.\ \mathcal{P}\ \varphi) \leq \mathcal{P}\ \psi$
      using
        ‹*?rhs*›
        exclusive-sum-list-identity
        monotonicity
      by *fastforce*
  qed
qed

lemma (**in** *classical-logic*) *dirac-inequality-completeness*:
  $(\forall\ \delta \in dirac\text{-}measures.\ \delta\ \varphi \leq \delta\ \psi) = \vdash \varphi \to \psi$
proof −
  have $\vdash \coprod\ [\varphi]$
    by (*simp add*: *conjunction-right-elimination negation-def*)
  hence $(\vdash \coprod\ [\varphi] \wedge\ \vdash \bigsqcup\ [\varphi] \to \psi) = \vdash \varphi \to \psi$
    by (*metis*
        arbitrary-disjunction.simps(1)
        arbitrary-disjunction.simps(2)
        disjunction-def implication-equivalence
        negation-def
        weak-biconditional-weaken)
  thus *?thesis*
    using *dirac-exclusive-implication-completeness* [**where** $\Phi=[\varphi]$]
    by *auto*
qed

2.4 Implication Inequality Completeness

The following theorem establishes the converse of $\vdash \varphi \to \psi \Longrightarrow \mathcal{P}\ \varphi \leq \mathcal{P}\ \psi$, which was proved in §1.4.

theorem (**in** *classical-logic*) *implication-inequality-completeness*:
  $(\forall\ \mathcal{P} \in \mathit{probabilities}.\ \mathcal{P}\ \varphi \leq \mathcal{P}\ \psi) = \vdash \varphi \to \psi$
proof −
  have $\vdash \coprod\ [\varphi]$
    by (*simp add*: *conjunction-right-elimination negation-def*)
  hence $(\vdash \coprod\ [\varphi] \wedge\ \vdash \bigsqcup\ [\varphi] \to \psi) = \vdash \varphi \to \psi$
    by (*metis*
        arbitrary-disjunction.simps(*1*)
        arbitrary-disjunction.simps(*2*)
        disjunction-def implication-equivalence
        negation-def
        weak-biconditional-weaken)
  thus *?thesis*
    using *exclusive-implication-completeness* [**where** Φ=[φ]]
    by *simp*
qed

2.5 Characterizing Logical Exclusiveness In Probability Logic

Finally, we can say that $\mathcal{P}\ (\bigsqcup\ \Phi) = (\sum \varphi{\leftarrow}\Phi.\ \mathcal{P}\ \varphi)$ if and only if the propositions in Φ are mutually exclusive (i.e. $\vdash \coprod\ \Phi$). This result also obtains for sets.

lemma (**in** *classical-logic*) *dirac-exclusive-list-summation-completeness*:
  $(\forall\ \delta \in \mathit{dirac\text{-}measures}.\ \delta\ (\bigsqcup\ \Phi) = (\sum \varphi{\leftarrow}\Phi.\ \delta\ \varphi)) = \vdash \coprod\ \Phi$
  by (*metis*
      antisym-conv
      dirac-exclusive-implication-completeness
      dirac-list-summation-completeness
      trivial-implication)

theorem (**in** *classical-logic*) *exclusive-list-summation-completeness*:
  $(\forall\ \mathcal{P} \in \mathit{probabilities}.\ \mathcal{P}\ (\bigsqcup\ \Phi) = (\sum \varphi{\leftarrow}\Phi.\ \mathcal{P}\ \varphi)) = \vdash \coprod\ \Phi$
  by (*metis*
      antisym-conv
      exclusive-implication-completeness
      list-summation-completeness
      trivial-implication)

lemma (**in** *classical-logic*) *dirac-exclusive-set-summation-completeness*:
  $(\forall\ \delta \in \mathit{dirac\text{-}measures}.\ \delta\ (\bigsqcup\ \Phi) = (\sum \varphi \in \mathit{set}\ \Phi.\ \delta\ \varphi))$
    $= \vdash \coprod\ (\mathit{remdups}\ \Phi)$
  by (*metis*

(*mono-tags*)
eq-iff
dirac-exclusive-implication-completeness
dirac-set-summation-completeness
trivial-implication
set-remdups
sum.set-conv-list *eq-iff*
dirac-exclusive-implication-completeness
dirac-set-summation-completeness
trivial-implication
set-remdups
sum.set-conv-list
antisym-conv)

theorem (**in** *classical-logic*) *exclusive-set-summation-completeness*:
$(\forall\ \mathcal{P} \in probabilities.$
$\mathcal{P}\ (\bigsqcup\ \Phi) = (\sum \varphi \in set\ \Phi.\ \mathcal{P}\ \varphi)) = \vdash \coprod\ (remdups\ \Phi)$
by (*metis*
(*mono-tags*, *opaque-lifting*)
antisym-conv
exclusive-list-summation-completeness
exclusive-implication-completeness
implication-inequality-completeness
set-summation-completeness
order.refl
sum.set-conv-list)

lemma (**in** *probability-logic*) *exclusive-list-set-inequality*:
assumes $\vdash \coprod\ \Phi$
shows $(\sum \varphi \leftarrow \Phi.\ \mathcal{P}\ \varphi) = (\sum \varphi \in set\ \Phi.\ \mathcal{P}\ \varphi)$
proof −
have *distinct* (*remdups* Φ) **using** *distinct-remdups* **by** *auto*
hence *duplicates* (*remdups* Φ) = {}
by (*induct* Φ, *simp*+)
moreover have *set* (*remdups* Φ) = *set* Φ
by (*induct* Φ, *simp*, *simp add*: *insert-absorb*)
moreover have $(\forall \varphi \in duplicates\ \Phi.\ \vdash \sim \varphi)$
$\wedge\ (\forall\ \varphi \in set\ \Phi.\ \forall\ \psi \in set\ \Phi.\ (\varphi \neq \psi) \longrightarrow \vdash \sim (\varphi \sqcap \psi))$
using
assms
exclusive-elimination1
exclusive-elimination2
set-deduction-base-theory
by *blast*
ultimately have
$(\forall \varphi \in duplicates\ (remdups\ \Phi).\ \vdash \sim \varphi)$
$\wedge\ (\forall\ \varphi \in set\ (remdups\ \Phi).\ \forall\ \psi \in set\ (remdups\ \Phi).$
$(\varphi \neq \psi) \longrightarrow \vdash \sim (\varphi \sqcap \psi))$
by *auto*

hence $\vdash \coprod$ (*remdups* Φ)
  by (*meson exclusive-equivalence set-deduction-base-theory*)
hence $(\sum \varphi \in set\ \Phi.\ \mathcal{P}\ \varphi) = \mathcal{P}\ (\bigsqcup\ \Phi)$
  by (*metis*
      arbitrary-disjunction-remdups
      biconditional-equivalence
      exclusive-sum-list-identity
      sum.set-conv-list)
moreover have $(\sum \varphi \leftarrow \Phi.\ \mathcal{P}\ \varphi) = \mathcal{P}\ (\bigsqcup\ \Phi)$
  by (*simp add*: *assms exclusive-sum-list-identity*)
ultimately show *?thesis* **by** *metis*
qed

unbundle *funcset-syntax*

end

Bibliography

[1] G. Boole. Chapter XVI. On The Theory Of Probabilities. In *An Investigation of the Laws of Thought On Which Are Founded the Mathematical Theories of Logic and Probabilities*, pages 243–252. 1853.

[2] B. De Finetti. Sui passaggi al limite nel calcolo delle probabilità. *Reale Istituto Lombardo di Scienze e Lettere*, 63:1–12, 1930.

[3] M. Eberl. Buffon's needle problem. *Archive of Formal Proofs*, June 2017.

[4] B. R. Gaines. Fuzzy and probability uncertainty logics. *Information and Control*, 38(2):154–169, Aug. 1978.

[5] G. Gerla. Inferences in probability logic. *Artificial Intelligence*, 70(1-2):33–52, Oct. 1994.

[6] T. Hailperin. Probability Logic. *Notre Dame Journal of Formal Logic*, 25(3):198–212, July 1984.

[7] T. Hailperin. *Boole's Logic and Probability: A Critical Exposition from the Standpoint of Contemporary Algebra, Logic and Probability Theory.* Number 85 in Studies in Logic and the Foundations of Mathematics. North-Holland, Amsterdam, 2. ed, rev. and enl edition, 1986.

[8] T. Hailperin. *Sentential Probability Logic: Origins, Development, Current Status, and Technical Applications.* Lehigh University Press, Bethlehem : London ; Cranbury, N.J, 1996.

[9] A. Horn and A. Tarski. Measures in Boolean algebras. *Transactions of the American Mathematical Society*, 64(3):467–467, Mar. 1948.

[10] A. Kolmogoroff. Chapter 1. Die elementare Wahrscheinlichkeitsrechnung. In *Grundbegriffe der Wahrscheinlichkeitsrechnung*, number 2 in Ergebnisse der Mathematik und Ihrer Grenzgebiete, pages 1–12. Springer-Verlag Berlin Heidelberg, first edition, 1933.

[11] N. Rescher. *Many-Valued Logic.* McGraw-Hill, New York, first edition, Jan. 1969.

[12] L. J. Savage. Difficulties in the Theory of Personal Probability. *Philosophy of Science*, 34(4):305–310, 1967.

[13] P. Suppes. Probabilistic Inference and the Concept of Total Evidence. In J. Hintikka and P. Suppes, editors, *Studies in Logic and the Foundations of Mathematics*, volume 43 of *Aspects of Inductive Logic*, pages 49–65. Elsevier, Jan. 1966.

[14] B. Weatherson. From Classical to Intuitionistic Probability. *Notre Dame Journal of Formal Logic*, 44(2):111–123, Apr. 2003.